



SPSS Tip 11.2

Using syntax to recode

The syntax file, **RecodeGlastonburyData.sps**, creates all the dummy variables we've discussed. Load this file or type these commands into a new syntax window (see Section 4.10):

```
DO IF(1-MISSING(change)).  
RECODE music (3=1)(ELSE = 0) INTO Crusty.  
RECODE music (2=1)(ELSE = 0) INTO Metaller.  
RECODE music (1=1)(ELSE = 0) INTO Indie_Kid.  
END IF.  
VARIABLE LABELS Crusty 'No Affiliation vs. Crusty'.  
VARIABLE LABELS Metaller 'No Affiliation vs. Metaller'.  
VARIABLE LABELS Indie_Kid 'No Affiliation vs. Indie Kid'.  
VARIABLE LEVEL Crusty Metaller Indie_Kid (Nominal).  
FORMATS Crusty Metaller Indie_Kid (F1.0).  
EXECUTE.
```

Each *recode* command does the equivalent of the dialog box in Figure 11.15. So, the three lines beginning *recode* create three new variables (**Crusty**, **Metaller** and **Indie_Kid**), based on the variable **music**. The first variable (**Crusty**) takes on values of 1 if **music** is 3 and 0 for all other values. The second is coded 1 if **music** is 2 and 0 for all other values, and so on for the third dummy variable. These *recode* commands are within an *if* statement (beginning *do if* and ending with *end if*), which means that they will only be executed if a certain condition is met. The condition we have set is $1 - \text{MISSING}(\text{change})$, which is the same as in Figure 11.14 (see main text).

The *variable labels* command assigns the text in the quotations as labels for the variables **Crusty**, **Metaller**, and **Indie_Kid** respectively. The *variable level* command then sets these three variables to be 'nominal', and the *formats* command changes the variables to have a width of 1 and 0 decimal places (hence the 1.0). The *execute* command executes the commands above (without it nothing works). Note that every line ends with a full stop.

